「情報システムの概要」

教科「情報」現職教員等講習会資料

平成13年8月1日（水）

gbb60166@gmail.com
■　目的

情報システムを開発するために必要となる設計の重要性について理解すると共に，設計に関する基本的な知識や技術を習得し，システム開発者として実際に活用する能力と態度を身に付ける。

■　内容

情報システムを開発するにはコンピュータに関する知識だけではなく，問題を発見したり，発見した問題をどのように解決するかという手順・アルゴリズムを正確に理解する能力が求められる。

しかし実際に運用してみないと発見されない問題も多いことも事実である。そのためシステム開発者はトータル・コーディネータとして，システムの保守・点検にあたることが求められる。

本講座では実際に実務で使用されている学校給食栄養計算ソフトを例に取り，「システム設計」「コーディング」「テスト」「ドキュメント・ヘルプファイルの作成」「アーカイブの作成」「配布」「バッチファイルの使用」「セキュリティ」「著作権」「アルゴリズム」等について説明する。

さらに Excel VBA (Visual Basic for Applications) を使用して，簡単なシステム開発を行う。

■　目標

入試の資料作成用ユーザーフォームを作る。

ユーザーフォームを作るためにはVisual Basic Editor(VBE)を使用します。VBEを呼び出すためのボタンをメニューバーに表示することにしましょう。

　メニューバー上にマウスカーソルを持っていき右クリックします。すると表示することができる項目の一覧が表示されるので「Visual Basic」にチェックを付けます。

するとコマンドバーが表示されます。

をクリックすると「Visual Basic Editor」が表示されます。

ユーザーフォームを作成するには[image: image5.png]


をクリックします。

ユーザーフォーム上に「Label」「CommandButton」などを配置して「Caption」をわかりやすい名前に変更します。

	コントロール名
	Caption
	備考
	対応するコントロール

	Label1
	受検番号
	
	Textbox1

	Label2
	氏名
	
	Textbox2

	Label3
	ふりがな
	
	Textbox3

	Label4
	中学校名
	
	ComboBox1

	Label5
	卒業年度
	
	ComboBox2

	Frame1
	性別
	
	OptionButton1, OptionButton2

	OptionButton1
	男
	
	Frame1

	OptionButton2
	女
	
	Frame1

	Frame2
	学力検査点数
	
	Label6～11,TextBox4～9

	Label6
	国語
	
	TextBox4

	Label7
	理科
	
	TextBox5

	Label8
	英語
	
	TextBox6

	Label9
	社会
	
	TextBox7

	Label10
	数学
	
	TextBox8

	Label11
	合計
	
	TextBox9

	TextBox1
	
	受検番号
	Label1

	TextBox2
	
	氏名
	Label2

	TextBox3
	
	ふりがな
	Label3

	TextBox4
	
	国語の点数
	Label6

	TextBox5
	
	理科の点数
	Label7

	TextBox6
	
	英語の点数
	Label8

	TextBox7
	
	社会の点数
	Label9

	TextBox8
	
	数学の点数
	Label10

	TextBox9
	
	合計の点数
	Label11

	ComboBox1
	
	中学校名
	Label4

	ComboBox2
	
	卒業年度
	Label5

	CommandButton1
	前の受検番号
	
	

	CommandButton2
	次の受検番号
	
	

	CommandButton3
	閉じる
	
	

	
	
	
	


「書式」「整列」，「同じサイズを揃える」「間隔を均等にする」などを使い整形する。

「プロパティウィンドウ」の数値を直接書き換えても良い。

[Ctrl]を押しながらクリックをしていくと不連続なものを選択できる

[Shift]を押しながら，最初と最後のものをクリックすると連続したものを選択出来る

[Ctrl]を押しながらコントロールをドラッグするとコントロールのコピーが作成できる

　ユーザーフォームを表示

　コードの表示

■テスト用の数値を代入してみる

Private Sub CommandButton3_Click()

    Unload Me

End Sub

Private Sub UserForm_Initialize()

    TextBox1 = 101

    TextBox2 = "佐賀　太郎"

    TextBox3 = "さが　たろう"

    OptionButton1 = True

    ComboBox1 = "第一"

    ComboBox2 = "H11.3"

    TextBox4 = 23

    TextBox5 = 41

    TextBox6 = 12

    TextBox7 = 7

    TextBox8 = 0

    TextBox9 = CInt(TextBox4) + CInt(TextBox5) + _

        CInt(TextBox6) + CInt(TextBox7) + CInt(TextBox8)

End Sub
※「半角スペースと_（アンダースコア）」は「次の行に続く」というマークです。「CInt」は整数に変換する関数です

ボタンをクリックすると作成したフォームが表示されます

講義1st.xls

■「合計欄」はロックした方が良い。合計点数は自動計算できるのでユーザーが書き換える必要はない。

→プロパティウィンドウを使い，「Locked」プロパティを「True」にする

■問題点　「中学校名」「卒業年度」はロックした方が良い。「第一」と入力されたり「第１」と入力されるおそれがある。

→ComboBox1，ComboBox2の「Style」を「fmStyleDropDownList」に変更する。

· 国語などの点数を書き換えたとき，合計点数が自動的に変更されるようにしたい

→下記のコードを追加

Private Sub TextBox4_AfterUpdate()

    TextBox9 = CInt(TextBox4) + CInt(TextBox5) + _

        CInt(TextBox6) + CInt(TextBox7) + CInt(TextBox8)

End Sub

Private Sub TextBox5_AfterUpdate()

    TextBox9 = CInt(TextBox4) + CInt(TextBox5) + _

        CInt(TextBox6) + CInt(TextBox7) + CInt(TextBox8)

End Sub

Private Sub TextBox6_AfterUpdate()

    TextBox9 = CInt(TextBox4) + CInt(TextBox5) + _

        CInt(TextBox6) + CInt(TextBox7) + CInt(TextBox8)

End Sub

Private Sub TextBox7_AfterUpdate()

    TextBox9 = CInt(TextBox4) + CInt(TextBox5) + _

        CInt(TextBox6) + CInt(TextBox7) + CInt(TextBox8)

End Sub

Private Sub TextBox8_AfterUpdate()

    TextBox9 = CInt(TextBox4) + CInt(TextBox5) + _

        CInt(TextBox6) + CInt(TextBox7) + CInt(TextBox8)

End Sub

■「受検番号」「国語」～「数学」は数値しか入力されないので日本語入力を不可能にする。

→「IMEMode」を「fmIMEModedisable」にする

· 「氏名」「ふりがな」は日本語入力をするので自動的に日本語入力が出来る状態にしたい。

→「IMEMode」を「fmIMEModeOn」にする

■中学校名を選択出来るようにしたい

中学校名をシートに書き込んでおけば良いようにする

「Sheet2」のセルA1からセルA5に中学校名を入力しておく。次のコードを追加する（Dim 中学校名 As Range は，かならず一行目に書いて下さい）

A1：第一

A2：第五

A3：西唐津

A4：鏡

A5：厳木

Dim 中学校名 As Range

Private Sub UserForm_Initialize()　の中に追加する

    Set 中学校名 = Worksheets("Sheet2").Range("A1").CurrentRegion

    ComboBox1.RowSource = "Sheet2!" & 中学校名.Address

※CurrentRegionは連続した範囲を選択するプロパティです。CurrentRegion を使うことで，中学校名を増やしてもコードを変更する必要がなくなります。

※Addressは範囲のアドレスを「A1:A5」のように返すプロパティです。

■卒業年度を選択出来るようにしたい

「Sheet2」のセルC1からセルC5に卒業年度を入力しておく。

C1 : H14.3

C2 : H13.3

C3 : H12.3

C4 : H11.3

C5 : H10.3

そして次のコードを追加する。（Dim 卒業年度 As Range は，Dim 中学校名 As Range の次の行に書いて下さい）

Dim 卒業年度 As Range

Private Sub UserForm_Initialize()　の中に追加する

    Set 卒業年度 = Worksheets("Sheet2").Range("C1").CurrentRegion

    ComboBox2.RowSource = "Sheet2!" & 卒業年度.Address

講義2nd.xls

· エラーチェック機能の追加（点数は50点以下）

Private Sub TextBox4_AfterUpdate()　の中に追加

If TextBox4 > 50 Then

    MsgBox "点数オーバー"

    Exit Sub

End If

TextBox5～TextBox8も同様に書き直す。

※MsgBox はメッセージを伝えるダイアログを表示する関数です。

· ユーザがフォームを呼び出せるようにする。

「標準モジュール」追加する。標準モジュール「Module1」に次のコードを追加する。

Sub フォーム表示()

    UserForm1.Show

End Sub

するとExcelのメニューから「ツール(T)」「マクロ(M)」「マクロ(M)」でユーザーフォームを呼び出せるようになります。

講義3rd.xls

· ダミーデータをシートに書く

ダミーのデータをシートに書き込み，フォームを初期化するときにシートにあるデータをフォームに転記するように変更する。

Private Sub UserForm_Initialize()　内を変更する

    Worksheets("Sheet1").Select

    i = 1

    TextBox1 = Cells(i + 1, 1)

    TextBox2 = Cells(i + 1, 2)

    TextBox3 = Cells(i + 1, 3)

    If Cells(i + 1, 4) = "女" Then

        OptionButton2 = True

    Else

        OptionButton1 = True

    End If

    ComboBox1 = Cells(i + 1, 5)

    ComboBox2 = Cells(i + 1, 6)

    Set 中学校名 = Worksheets("Sheet2").Range("A1").CurrentRegion

    ComboBox1.RowSource = "Sheet2!" & 中学校名.Address

    Set 卒業年度 = Worksheets("Sheet2").Range("C1").CurrentRegion

    ComboBox2.RowSource = "Sheet2!" & 卒業年度.Address

    TextBox4 = Cells(i + 1, 7)

    TextBox5 = Cells(i + 1, 8)

    TextBox6 = Cells(i + 1, 9)

    TextBox7 = Cells(i + 1, 10)

    TextBox8 = Cells(i + 1, 11)

    TextBox9 = CInt(TextBox4) + CInt(TextBox5) + _

        CInt(TextBox6) + CInt(TextBox7) + CInt(TextBox8)

End Sub
※Cells(2,5)は「2行5列」（つまりセルE2に相当）します。

· タブオーダーの設定

ユーザーフォームはタブオーダーを設定した方が使い勝手が良くなります。メニューバーから「表示(V)」「タブオーダー(A)」の順に選択してオーダー（順番）を設定しましょう。

講義4th.xls

■「次の受検番号」ボタンを押したときに次の行に移るようにする

CInt(TextBox4) をCInt(Val(TextBox4))に変更（空欄のとき不具合が起こるため）

※Valは文字を数値に変換する関数です。（空白文字列””は0として扱われます）

次のコードを追加

Dim i As Integer　を先頭行に記入
Private Sub CommandButton2_Click()

    i = i + 1

    TextBox1 = Cells(i + 1, 1)
    TextBox2 = Cells(i + 1, 2)

    TextBox3 = Cells(i + 1, 3)

    If Cells(i + 1, 4) = "女" Then

        OptionButton2 = True

    Else

        OptionButton1 = True

    End If

    ComboBox1 = Cells(i + 1, 5)
    ComboBox2 = Cells(i + 1, 6)

    TextBox4 = Cells(i + 1, 7)

    TextBox5 = Cells(i + 1, 8)

    TextBox6 = Cells(i + 1, 9)

    TextBox7 = Cells(i + 1, 10)

    TextBox8 = Cells(i + 1, 11)

    TextBox9 = CInt(Val(TextBox4)) + CInt(Val(TextBox5)) + _

    CInt(Val(TextBox6)) + CInt(Val(TextBox7)) + CInt(Val(TextBox8))

End Sub
講義5th.xls

· これだけではシートへの転記が行われていないため，ユーザーフォームを書き直してもシートへ反映されない。シートへ転記するためのコードを追加する。タイミングは「次の受検番号」ボタンを押したときである。

· ｉの意味が分かりにくいので変更する。シートの一行目は項目名が記入されているため。一番目のデータは二行目にある。そのためずれが生じて分かりにくい。データ範囲を定義してｉとデータが格納されている行数を一致させる。TextBoxの数値をシートに転記しても数値として認識されていない。.Valueを付加する。

Dim データ範囲 As Range

Private Sub CommandButton2_Click()　を変更

    データ範囲.Cells(i, 1) = TextBox1.Value

    データ範囲.Cells(i, 2) = TextBox2.Value

    データ範囲.Cells(i, 3) = TextBox3.Value

    If OptionButton1 = True Then データ範囲.Cells(i, 4) = "男"

    If OptionButton2 = True Then データ範囲.Cells(i, 4) = "女"

    データ範囲.Cells(i, 5) = ComboBox1.Value

    データ範囲.Cells(i, 6) = ComboBox2.Value

    データ範囲.Cells(i, 7) = TextBox4.Value

    データ範囲.Cells(i, 8) = TextBox5.Value

    データ範囲.Cells(i, 9) = TextBox6.Value

    データ範囲.Cells(i, 10) = TextBox7.Value

    データ範囲.Cells(i, 11) = TextBox8.Value

    データ範囲.Cells(i, 12) = TextBox9.Value

    i = i + 1

    TextBox1 = i
    TextBox2 = データ範囲.Cells(i, 2)

    TextBox3 = データ範囲.Cells(i, 3)

    If データ範囲.Cells(i, 4) = "女" Then

        OptionButton2 = True

    Else

        OptionButton1 = True

    End If

    ComboBox1 = データ範囲.Cells(i, 5)

    ComboBox2 = データ範囲.Cells(i, 6)

    TextBox4 = データ範囲.Cells(i, 7)

    TextBox5 = データ範囲.Cells(i, 8)

    TextBox6 = データ範囲.Cells(i, 9)

    TextBox7 = データ範囲.Cells(i, 10)

    TextBox8 = データ範囲.Cells(i, 11)

    TextBox9 = CInt(Val(TextBox4)) + CInt(Val(TextBox5)) + _

    CInt(Val(TextBox6)) + CInt(Val(TextBox7)) + CInt(Val(TextBox8))

End Sub
Private Sub UserForm_Initialize()　内を変更する

    Worksheets("Sheet1").Select

    Set データ範囲 = Worksheets("Sheet1").Range("A2:L500")

    i = 1

    TextBox1 = i 
    TextBox2 = データ範囲.Cells(i, 2)

    TextBox3 = データ範囲.Cells(i, 3)

    If データ範囲.Cells(i, 4) = "女" Then

        OptionButton2 = True

    Else

        OptionButton1 = True

    End If

    ComboBox1 = データ範囲.Cells(i, 5)

    ComboBox2 = データ範囲.Cells(i, 6)

    Set 中学校名 = Worksheets("Sheet2").Range("A1").CurrentRegion

    ComboBox1.RowSource = "Sheet2!" & 中学校名.Address

    Set 卒業年度 = Worksheets("Sheet2").Range("C1").CurrentRegion

    ComboBox2.RowSource = "Sheet2!" & 卒業年度.Address

    TextBox4 = データ範囲.Cells(i, 7)

    TextBox5 = データ範囲.Cells(i, 8)

    TextBox6 = データ範囲.Cells(i, 9)

    TextBox7 = データ範囲.Cells(i, 10)

    TextBox8 = データ範囲.Cells(i, 11)

    TextBox9 = CInt(Val(TextBox4)) + CInt(Val(TextBox5)) + _

        CInt(Val(TextBox6)) + CInt(Val(TextBox7)) + CInt(Val(TextBox8))

End Sub
講義6th.xls

· 受検しなかった科目のある生徒の点数を「空欄」にして，「閉じる」ボタンを押すと「点数オーバー」のメッセージが出るので改良をする。>50のチェックをするのは，点数が数値であるときだけにする。

If IsNumeric(TextBox4) And TextBox4 > 50 Then

    MsgBox "点数オーバー"

    Exit Sub

End If

TextBox5～TextBox8も同様

■受検番号を自動的に生成するのはやめにして，受検番号はあらかじめシートに書き込んでおくように変更する。実際の受検番号は男女で別にするのが普通であるため自動生成は難しい。

A2:A17に番号101～116を記入，A18:A25に番号501～508を記入する。

TextBox1がらみをすべて変更する。具体的に言うと

TextBox1 = i　を

TextBox1 = データ範囲.Cells(i, 1)　に書き換える。

■受検番号が501になったときには，自動的に「女」に変わるようにしたい。

Private Sub CommandButton2_Click()　内を変更する。

If Cells(i + 1, 4) = "女" Then

    OptionButton2 = True

Else

    OptionButton1 = True

End If
の部分を消去して，次のように書き直す

    If データ範囲.Cells(i, 1) > 500 Then

        OptionButton2 = True

    Else

        OptionButton1 = True

    End If

講義7th.xls

· タブオーダーを変更

合計の欄には[Tab]を押しても止まらないようにする。

プロパティウィンドウの「TabStop」を「False」にする。

数学の次のタブオーダーを「次の受検番号」に変更する。ほとんどの場合は次の番号に進むため，使い勝手が良くなるように改造。

→タブオーダーの変更

· 「卒業年度」のところに「空白」を許可したい。

ほとんどの生徒は新卒であるために，卒業年度を記入するのは過年度生だけにしたい。

中学校名はC2:C:6に移動してC1は空欄とする。

Private Sub UserForm_Initialize()　内を変更する
    Set 卒業年度 = Worksheets("Sheet2").Range("C1:C2").CurrentRegion

シートの新卒生徒の卒業年度「H14.3」は消去する。

· 「閉じる」ボタンを押したときデータを書き込むように変更したい

現在の使用では「閉じる」ボタンを押しても何もせずにフォームを閉じているので，データを書き込んで「閉じる」ボタンを押したときデータがシートに転記されない。

Private Sub CommandButton2_Click()　内を変更する

    データ転記

    i = i + 1

    TextBox1 = データ範囲.Cells(i, 1)

    TextBox2 = データ範囲.Cells(i, 2)

    TextBox3 = データ範囲.Cells(i, 3)

Private Sub CommandButton3_Click()　を変更する

    データ転記

    Unload Me

End Sub

下記のコードは複数のサブルーチンから呼び出されるので，同じコードを何度も書かなくて済むように，独立させる。こうすることでコードの保守性が高まる。

Sub データ転記()

    'データ範囲.Cells(i, 1) = TextBox1.Value

    データ範囲.Cells(i, 2) = TextBox2.Value

    データ範囲.Cells(i, 3) = TextBox3.Value

    If OptionButton1 = True Then データ範囲.Cells(i, 4) = "男"

    If OptionButton2 = True Then データ範囲.Cells(i, 4) = "女"

    データ範囲.Cells(i, 5) = ComboBox1.Value

    データ範囲.Cells(i, 6) = ComboBox2.Value

    データ範囲.Cells(i, 7) = TextBox4.Value

    データ範囲.Cells(i, 8) = TextBox5.Value

    データ範囲.Cells(i, 9) = TextBox6.Value

    データ範囲.Cells(i, 10) = TextBox7.Value

    データ範囲.Cells(i, 11) = TextBox8.Value

    データ範囲.Cells(i, 12) = TextBox9.Value

End Sub
講義8th.xls

■「前の受検番号」ボタンを押したときに前の行に移るようにする

コードとして独立させる

Sub データ代入()

    TextBox1 = データ範囲.Cells(i, 1)

    TextBox2 = データ範囲.Cells(i, 2)

    TextBox3 = データ範囲.Cells(i, 3)

    If データ範囲.Cells(i, 1) > 500 Then

        OptionButton2 = True

    Else

        OptionButton1 = True

    End If

    ComboBox1 = データ範囲.Cells(i, 5).Value

    ComboBox2 = データ範囲.Cells(i, 6)

    TextBox4 = データ範囲.Cells(i, 7)

    TextBox5 = データ範囲.Cells(i, 8)

    TextBox6 = データ範囲.Cells(i, 9)

    TextBox7 = データ範囲.Cells(i, 10)

    TextBox8 = データ範囲.Cells(i, 11)

    TextBox9 = CInt(Val(TextBox4)) + CInt(Val(TextBox5)) + _

    CInt(Val(TextBox6)) + CInt(Val(TextBox7)) + CInt(Val(TextBox8))

End Sub

Private Sub CommandButton1_Click()　を新規に追加

    データ転記

    i = i - 1

    データ代入

End Sub

Private Sub CommandButton2_Click()　内を変更する

    データ転記

    i = i + 1

    データ代入

End Sub
■受検番号が101のとき「前の受検番号」ボタンを押すと，データが無いのでエラーで止まる。

Private Sub CommandButton1_Click()

    データ転記

    If i >= 2 Then

        i = i - 1

        データ代入

    Else

        MsgBox "前のデータは存在しません。"

    End If

End Sub

■受検番号を書き換えたときその番号の生徒のデータが表示されるようにしたい

この要求は結構難しい。入力された受検番号を検索して何行目にあるのか調べることにする。

Private Sub TextBox1_AfterUpdate()

    検索値 = Application.Match(Val(TextBox1), データ範囲.Columns(1), 0)

    データ転記

    i = 検索値

    データ代入

End Sub

■存在しない受検番号が入力されたら警告を出し，実行を中止したい。

Private Sub TextBox1_AfterUpdate()

    If IsError(Application.Match(Val(TextBox1), データ範囲.Columns(1), 0)) Then

        MsgBox "入力された受検番号は存在しません。", vbCritical

        TextBox1 = データ範囲.Cells(i, 1)

        Exit Sub

    End If

    検索値 = Application.Match(Val(TextBox1), データ範囲.Columns(1), 0)

    データ転記

    i = 検索値

    データ代入

End Sub

· セットフォーカス

フォームが表示されたときはTextBox2（氏名欄）にフォーカスを当てたい。こうすることでマウスでクリックすることなくスムーズにデータ入力が出来る。

Private Sub UserForm_Initialize()

Private Sub CommandButton1_Click()

Private Sub CommandButton2_Click()

Private Sub TextBox1_AfterUpdate()

の　End Sub　の直前に次のステートメントを追加する

    TextBox2.SetFocus

■「次の受検番号」ボタンを押し続けると「性別」と「合計点数」記入されて格好悪い。コードを修正する。

Sub データ転記()

If Not TextBox2 = "" Then

    'データ範囲.Cells(i, 1) = TextBox1.Value

    データ範囲.Cells(i, 2) = TextBox2.Value

    データ範囲.Cells(i, 3) = TextBox3.Value

    If OptionButton1 = True Then データ範囲.Cells(i, 4) = "男"

    If OptionButton2 = True Then データ範囲.Cells(i, 4) = "女"

    データ範囲.Cells(i, 5) = ComboBox1.Value

    データ範囲.Cells(i, 6) = ComboBox2.Value

    データ範囲.Cells(i, 7) = TextBox4.Value

    データ範囲.Cells(i, 8) = TextBox5.Value

    データ範囲.Cells(i, 9) = TextBox6.Value

    データ範囲.Cells(i, 10) = TextBox7.Value

    データ範囲.Cells(i, 11) = TextBox8.Value

    データ範囲.Cells(i, 12) = TextBox9.Value

End If

End Sub

· ここをこう改善したい

· 「氏名」と「ふりがな」欄は，姓と名の間が「全角スペース一つ」で区切られているかチェックする

· 直接シートを編集することができないようにしたい。

· 「ふりがな」は「ひらがなだけ」になっているかチェックする

· 点数の所に数字以外が入力出来ないようにする。（アルファベットや＋－＊／の入力が出来ないようにしたい）

